

Progettiamo un circuito automobilistico

In questo numero tratteremo un solo argomento, relativamente semplice, nell'intento di "catturare" quei nuovi lettori che si vogliono accostare alla grafica con il computer, ma vogliono partire da tematiche semplici e comprensibili, e soprattutto svilupparli su tutti i microcomputer dotati di un minimo di funzioni grafiche.

L'argomento che abbiamo scelto è il disegno di circuiti automobilistici.

Anche questa volta non presentiamo un programma "usa e getta", cioè da copiare e usare, ma suggeriamo una tematica che può essere ampliata e sviluppata per esempio realizzando, sulla base del circuito tracciato, dei Game di Formula 1, bi o tridimensionali.

L'idea che vogliamo sviluppare è quella di realizzare un circuito automobilistico chiuso, ovvero vogliamo realizzare una successione di curve e di rettilinei, in modo che il percorso non abbia sovrapposizioni e che, ad un certo punto si chiuda. Il risultato è quello illustrato nella figura 1, che è l'output sul display a cristalli liquidi di un microcomputer portatile CASIO FP 200.

Qualsiasi tema di Computer Grafica, anche il più semplice, necessita di uno studio preliminare a "tavolino", per determinare correttamente tutti gli elementi del problema e per distinguere quelli che vanno dati inizialmente e quelli, che, invece, vanno calcolati via via nel corso dell'esecuzione. Il risultato dello studio sul tema del circuito automobilistico è riportato in figura 2.

Un po' di teoria

Supponiamo di voler tracciare un segmento su un foglio di carta. Gli elementi che occorrono sono:

punto iniziale $P_0 (X_0, Y_0)$
 angolo iniziale A_0
 lunghezza segmento L_0
 oppure

punto iniziale $P_0 (X_0, Y_0)$
 punto finale $P_1 (X_1, Y_1)$

Da un punto di vista matematico i due casi sono equivalenti in quanto dato punto iniziale, angolo e lunghezza, si trova con una semplice formula trigonometrica il punto finale.

Quindi per ogni rettilineo del nostro circuito occorre avere il punto e l'angolo iniziali e la sua lunghezza.

Ma questi quattro dati sono necessari solo per il primo rettilineo perché per i successivi il punto iniziale e la sua inclinazione derivano da tutti i rettilinei e le curve

tracciate in precedenza. Per i rettilinei successivi al primo, dunque, l'unico dato da fornire è la lunghezza.

Per quanto riguarda le curve, le supporremo circolari. Sarà peraltro possibile implementare il programma prevedendo altri tipi di curve (parabole, spirali, ecc.).

Se le curve sono circolari i dati da fornire sono solo due: R, A (raggio della curva e suo sviluppo angolare). Tramite questi due dati, utilizzando le formule trigonometriche di figura 2, si ottengono: innanzitutto il centro della curva, ricavato essendo noto il punto iniziale della curva, (corrispondente al punto finale del rettilineo), e tracciando la perpendicolare al rettilineo stesso passante da questo punto e prendendo su di

essa una distanza pari al raggio. Noto il centro della curva e il suo sviluppo angolare si trova facilmente il punto finale. Quindi i vari punti iniziali e finali di rettilinei e curve si calcolano via via, mentre l'angolo di inclinazione va incrementato dopo ogni curva, dell'angolo della curva stessa.

Problemi di chiusura

Il circuito deve chiudere.

Da un punto di vista matematico vuol dire che la somma di tutti gli angoli deve dare comunque un angolo giro. E vuol dire che il punto finale dell'ultima curva deve coincidere con il punto iniziale del tracciato.

Questo controllo, in fase di impostazione dei dati, lo può eseguire il calcolatore. Il controllo, invece, che non vi siano sovrapposizioni del percorso lo eseguiremo "a




Figura 1 - Output del programma circuito-1. È ottenuto sul display a cristalli liquidi del Computer Portatile Casio FP 200.




Figura 2 - Disegno e formule. Prima di affrontare il programma è bene approfondire il problema dal punto di vista matematico.

vista". La sua esecuzione a programma rallenterebbe non poco l'elaborazione.

In definitiva abbiamo stabilito cosa occorre per tracciare il circuito:

- inizialmente il punto di partenza e l'inclinazione del primo rettilineo;
- per ogni rettilineo una lunghezza;
- per ogni curva un raggio ed un angolo.

Poiché ogni rettilineo è separato da un altro da una curva e poiché ogni due curve successive sono separate da un rettilineo (al limite lungo θ) i dati che occorrono sono $3 * N + 3$, se N sono le curve del circuito.

Un'altra considerazione prima di passare ad esaminare i programmi. Poiché i punti sono calcolati in successione, partendo ogni volta da quello calcolato prima, eventuali imprecisioni nel calcolo si sommano ed è prevedibile che tali imprecisioni ci siano in quanto lavoriamo con funzioni trigonometriche e con angoli in radianti.

L'inconveniente viene in un certo senso attutito in quanto i punti, anche se calcolati in floating point, vengono alla fine tradotti in numeri interi per poter essere visualizzati, e quello che interessa ai fini pratici è proprio la corrispondenza del punto visualizzato e non di quello calcolato.

Il programma circuito 1

Il primo tracciato consiste nel disegno di un'unica linea (che potrebbe essere la mezzeria del circuito). I due programmi che vedremo poi prevedono invece le due linee, che rappresentano i lati della carreggiata e che si ottengono variando in più o in meno i raggi delle varie curve.

Il programma Circuito 1 è realizzato sul portatile CASIO FP 200 ed è listato in figura 3 mentre l'output è in figura 1.

I dati sono raggruppati all'inizio del programma e, a gruppi di tre, rappresentano la lunghezza del rettilineo, il raggio della curva successiva e il suo sviluppo in angoli sessagesimali.

Il problema della curva a sinistra o a destra è risolto imponendo il segno - alle curve antiorarie e con il riconoscimento di tale segno in fase di elaborazione.

Il valore 999 rappresenta l'END OF DATA, che fa finire il programma.

L'utilizzo di questo metodo per individuare la fine del programma è da preferire al metodo di stabilire in anticipo il numero dei dati, in quanto, in fase di test delle varie routine, basta infilare il valore 999 da qualche parte per ottenere la voluta elaborazione parziale.

In riga 24 c'è la pulizia delle variabili e dello schermo, nonché la definizione del primo punto P(X,Y) e dell'angolo di inclinazione del primo rettilineo.

Ci sono poi le due routine di tracciamen-

```

10 REM CIRCUITO-1
12 DATA 13,10,90,5,12,-120
14 DATA 20,10,-45,12,12,135
16 DATA 20,12,-120,12,7,-45
18 DATA 15,7,-75,25,10,45
20 DATA 2,10,-45,25,10,-75
22 DATA 6,12,120,15,13,-225,999
24 CLEAR:CLS:X=27:Y=34:A=0
26 QUAD(0,0)-(159,63):DRAW(X,Y)
28 REM RETTILINEO
30 READ L:IF L=999 THEN 50
32 X=X+L*COS(A):Y=Y+L*SIN(A)
34 DRAW-(X,Y)
36 REM CURVA
38 READ R,B:S=SGN(B):D=A+90
40 XC=X+R*COS(D)*S:YC=Y+R*SIN(D)*S
42 DRAW(XC,YC)
44 FOR C=A-90TOA+B-90STEP3*S
46 X=XC+R*COS(C)*S:Y=YC+R*SIN(C)*S
48 DRAW(X,Y):NEXT C:A=A+B:GOTO28
50 I$=INKEY$:IFI$=""THEN50ELSECLS:END
    
```

Figura 3.
Listato del programma circuito 1. I dati riferiti a rettilinei e curve sono raggruppati all'inizio del listato.

```

10 REM CIRCUITO-A
20 DATA 7,30,45,1,30,-45,20,20,-90,12,12,120
30 DATA 2,15,-150,10,14,180,60,18,-150,40,12,45
40 DATA 0,12,-45,35,20,-120,30,10,30,50,20,-45
50 DATA 20,22,135,10,12,-135,20,12,60
60 DATA 10,15,-60,10,15,-60,30,20,-45
70 DATA 0,20,60,21,5,11,-90
90 DATA 999,0,0
100 REM INIZIALIZZAZIONI
110 CLEAR:HOME:HGR2:HCOLDR=3:P=3:14159:Q=P/180
120 REM VALORE Z
130 IF Z THEN TEXT:HOME
140 M=49:N=152:A=0:X=M:Y=N:GOSUB 830:GOSUB 800
150 IF Z THEN PRINT "P. IN.":TAB(8)INT(X):TAB(16)INT(Y)
200 REM PRIMA LINEA
210 READ L,R,B:S=SGN(B):C=B*Q:IF L=999 THEN 400
220 X=X+L*COS(A):Y=Y+L*SIN(A):GOSUB 810
230 PP=PP+1:PRINT "P.":PP:TAB(8)INT(X):TAB(16)INT(Y)
240 D=A+P/2:E=R*S:J=X+E*COS(D)
250 K=Y+E*SIN(D):IF S=-1 THEN GOSUB 820
260 F=A-P/2:G=F+B*Q
270 IF Z THEN 300
280 FOR H=F TO G STEP Q*S*3
290 X=J+E*COS(H):Y=K+E*SIN(H):GOSUB 810:NEXT H
300 X=J+E*COS(G):Y=K+E*SIN(G):GOSUB 810
310 PP=PP+1:PRINT "P.":PP:TAB(8)INT(X)
320 PRINT TAB(16)INT(Y):TAB(22)"AG.":INT(A/Q+B+5)
330 IF S=-1 THEN GOSUB 820
340 A=A+B*Q:GOTO 200
400 REM SECONDA LINEA
410 RESTORE:X=M:Y=N-5:A=0:GOSUB 800:IF Z THEN END
420 READ L,R,B:S=SGN(B):C=B*Q:R=R+5*S
430 IF L=999 THEN END
440 X=X+L*COS(A):Y=Y+L*SIN(A):GOSUB 810
450 D=A+P/2:E=R*S:J=X+E*COS(D)
460 K=Y+E*SIN(D):IF S=1 THEN GOSUB 820
470 F=A-P/2:G=F+B*Q
480 FOR H=F TO G STEP Q*S*3
490 X=J+E*COS(H):Y=K+E*SIN(H):GOSUB 810:NEXT H
500 X=J+E*COS(G):Y=K+E*SIN(G):GOSUB 810
510 IF S=1 THEN GOSUB 820
520 A=A+B*Q:GOTO 420
800 HPLLOT X+.5,Y+.5:RETURN
810 HPLLOT TO X+.5,Y+.5:RETURN
820 HPLLOT J+.5,K+.5 TO X+.5,Y+.5:RETURN
830 HPLLOT 0,0 TO 279,0 TO 279,191 TO 0,191 TO 0,0
831 HPLLOT 2,2 TO 277,2 TO 277,189 TO 2,189 TO 2,2:RETURN
    
```

Figura 4 - Listato del programma circuito-A. Questa volta l'output è realizzato sul monitor grafico dell'Apple II, sul quale, data la sua definizione, si può ottenere anche il raddoppio delle linee.

to del rettilineo (righe 28-34) che avviene mediante l'individuazione del punto finale, dato il punto iniziale, e di tracciamento della curva (righe 36-48).

La curva viene tracciata individuando innanzitutto il suo centro, e per far questo occorre stabilire, tramite il segno dell'angolo, se la curva è oraria od antioraria.

Stabilito e disegnato il centro (PC (XC,YC) di riga 40), viene tracciata la curva vera e propria tramite un loop il cui step deve essere positivo o negativo a seconda che il senso di tracciamento sia orario o antiorario (riga 44).

La curva viene tracciata con l'istruzione DRAW, in quanto l'istruzione CIRCLE, che è più precisa, non è sempre presente nel soft di base dei vari microcomputer.

Alla fine si ritorna alla riga 28 e cioè al tracciamento del rettilineo.

Si esce da questo giro, come detto, quando viene letto il valore 999, e in questo caso si va alla riga 50. Con tale riga, lo diciamo ai neofiti, si evita che alla fine dell'esecuzione appaia un READY o qualcosa di simile sul video e sporchi il disegno, in quanto per finire il programma occorre premere un tasto qualsiasi.

Il programma circuito - A

Il secondo programma (listato di fig. 4) è simile al precedente ma presenta due implementazioni fondamentali.

La prima è il tanto atteso raddoppio della linea, rappresentante il circuito, e viene ottenuto raddoppiando le routine di disegno e differenziandole opportunamente.

La seconda è l'inserimento sulla funzione di PRINT, con la quale, ponendo il flag Z=1 in riga 120, il programma visualizza su monitor le successive coordinate dei punti calcolati e l'incremento dell'angolo.

L'output di tale opzione è in figura 5, dove si può notare che il punto iniziale corrisponde a quello finale e che il totale degli angoli con segno dà — 360 gradi, che cioè il circuito chiude perfettamente in senso antiorario.

Non essendo un programma che genera automaticamente il circuito, tale opzione di controllo è essenziale in fase di progettazione del tracciato.

La prima linea tracciata con la prima routine (200-340) rappresenta non la mezza del circuito ma una delle uno dei due bordi e si ottiene in modo analogo al programma precedente.

La seconda linea si ottiene invece spostando di un certo valore (5 pixel) il disegno. L'unica difficoltà consiste nell'individuare quando il raggio della curva va incrementato e quando va decrementato e a questo pensa, al solito, il valore S che indica il senso della curva (riga 420).

JRUN			
P. IN.	49	152	
P. 1	56	152	
P. 2	77	160	AG. 45
P. 3	77	161	
P. 4	99	170	AG. 0
P. 5	119	170	
P. 6	139	150	AG. -90
P. 7	139	138	
P. 8	157	127	AG. 30
P. 9	158	128	
P. 10	179	108	AG. -120
P. 11	174	99	
P. 12	198	85	AG. 60
P. 13	228	137	
P. 14	262	128	AG. -90
P. 15	262	88	
P. 16	265	80	AG. -45
P. 17	265	80	
P. 18	269	71	AG. -90
P. 19	269	36	
P. 20	239	19	AG. -210
P. 21	213	34	
P. 22	208	35	AG. -180
P. 23	158	35	
P. 24	144	41	AG. -225
P. 25	129	55	
P. 26	92	40	AG. -90
P. 27	92	30	
P. 28	71	21	AG. -225
P. 29	57	35	
P. 30	46	38	AG. -165
P. 31	36	36	
P. 32	22	40	AG. -225
P. 33	14	47	
P. 34	11	61	AG. -285
P. 35	18	90	
P. 36	28	102	AG. -330
P. 37	28	102	
P. 38	38	120	AG. -270
P. 39	38	141	
P. 40	49	152	AG. -360

Figura 5 - Output alfanumerico del programma circuito - A. È indispensabile in fase di progettazione un output alfanumerico per controllare la "chiusura del percorso".




Figura 6
Output
sul monitor
del programma
circuito - A.
La buona
definizione
grafica
del monitor
Apple II,
permette
di tracciare
con discreta
precisione
il circuito
con le sue
caratteristiche
geometriche.

Lo svolgimento è identico a quello visto in precedenza.

Si notino solo le due END. Quella di riga 410, che è la fine prevista in caso di opzione PRINT alfanumerico e che evita l'esecuzione della seconda elaborazione relativa alla seconda riga. L'altra END è quella dell'opzione grafica di riga 430. Qui il programma finisce quando viene letto il valore "tappo" 999.

Il programma circuito - 3

L'ultima variante del circuito prevede l'uscita grafica sul Printer Plotter Casio FP 1011, che abbiamo già utilizzato altre volte e il cui formato di uscita è adatto a questo tipo di programmi.

Ovviamente disponendo di plotter più grandi si possono prevedere uscite di dimensioni A4, A3, ma questo non aggiunge nulla al disegno, lo amplifica solamente.

Il listato è in figura 7 ed è simile ai due precedenti, salvo che contiene le routine di plottaggio richiamate via via durante l'esecuzione.

riga 74	accensione plotter
" 76	move della penna alzata
" 78	draw a penna abbassata
" 80	move e draw in successione
" 82	box

Anche in questo caso, e lo si può vedere dai due output presentati (figg. 8 e 9) viene individuato il centro di ciascuna curva e vengono tracciati i raggi iniziali e finali della curva stessa.

Il tracciamento di una curva in genere avviene tramite una apposita istruzione CIRCLE cui vanno passati i parametri coordinate del centro, raggio, angolo iniziale ed angolo finale.

Il soft di base del Casio FP 1011 prevede questa istruzione, ma noi abbiamo usato la

```

10 REM CIRCUITO-3
12 DATA 4,9,90,16,9,-120
14 DATA 10,8,-45,11,5,135
16 DATA 8,12,-120,15,8,-45
18 DATA 14,7,-75,11,10,45
20 DATA 1,10,-45,4,10,-75
22 DATA 6,10,120,12,10,-225,999
24 CLEAR:CLS:M=11:N=40:A=0
26 X=M:Y=N:GOSUB74:GOSUB82:GOSUB76
28 REM **** PRIMA LINEA
30 READL:IFL=999THEN48
32 X=X+L*COS(A):Y=Y+L*SIN(A):GOSUB78
34 READR,B:S=SGN(B):D=A+90
36 Q=R*S:J=X+Q*COS(D):K=Y+Q*SIN(D)
38 E=A-90:IFS=-1THENGOSUB80
40 FOR C=E TO E+B STEP S:X=J+Q*COS(C)
42 Y=K+Q*SIN(C):GOSUB78:NEXTC
44 IFS=-1THENGOSUB80
46 A=A+B:GOTO30

48 REM **** SECONDA LINEA
50 RESTORE:X=M:Y=N-3:A=0:GOSUB76
52 READL:IFL=999THEN70
54 X=X+L*COS(A):Y=Y+L*SIN(A):GOSUB78
56 READR,B:S=SGN(B):R=R+S*3:D=A+90
58 Q=R*S:J=X+Q*COS(D):K=Y+Q*SIN(D)
60 E=A-90:IFS=1THENGOSUB80
62 FOR C=E TO E+B STEP S:X=J+Q*COS(C)
64 Y=K+Q*SIN(C):GOSUB78:NEXTC
66 IFS=1THENGOSUB80
68 A=A+B:GOTO52
70 I$=INKEY$:IFI$=""THEN70ELSECLS:END
72 REM ROUTINE PLOT
74 LPRINTCHR$(28):CHR$(37):RETURN
76 LPRINT"M";X;" ";Y:RETURN
78 LPRINT"D,";X;" ";Y:RETURN
80 LPRINT"M"J","K:GOSUB78:RETURN
82 LPRINT"A0,0,96,80":RETURN

```

Figura 7 - Listato del programma circuito - 3. Questa volta l'uscita è sul printer plotter FP 1011 della CASIO.

istruzione di DRAW (HPLOT TO dell'Applesoft) che tutti i microcomputer grafici hanno.

In pratica si stabilisce un passo di incremento dell'angolo, si calcolano e si collegano tra di loro i vari punti creando una spezzata tanto più vicina alla curva quanto minore è lo step.

Diminuire lo step vuol dire però aumentare i tempi di esecuzione e nel caso di disegno su plotter c'è sempre il pericolo che la linea tracciata più lentamente risulti più marcata di quella tracciata più velocemente, e così in pratica, occorre trovare un giusto compromesso tra velocità e precisione.

Nei due esempi di output abbiamo posto lo step pari a 3 gradi (fig. 8) e questo come si vede ha comportato errori di chiusura nel caso che l'angolo della curva non fosse multiplo di 3. Nella figura 9 c'è il caso in cui lo step è pari all'intero angolo e questo equivale a collegare solo punto iniziale e punto finale della curva. **MC**




Figura 8 - Output del programma circuito - 3. L'errore di chiusura, come viene detto nel testo, dipende dall'eccessiva approssimazione usata nel calcolo degli angoli.




Figura 9 - Output del programma circuito - 3. Questo output è utile per testare la congruenza dei DATA e si ottiene ponendo lo STEP sull'angolo della curva pari all'intero angolo.

SICCOB 84

☉ PUBLICIS
A 343



LE DECISIONI DEL FUTURO SI PRENDONO A PARIGI

SALONE INTERNAZIONALE DELL'INFORMATICA, DELLA TELEMATICA,
COMUNICAZIONE, ORGANIZZAZIONE E AUTOMATIZZAZIONE DELL'UFFICIO

19-28 SETTEMBRE

CNIT PARIGI LA DÉFENSE

ESPOSITORI DI 28 PAESI

VISITATORI DI 115 PAESI

CONGRESSI-CONFERENZE

SERVIZIO ACCOGLIENZA PER I VISITATORI INTERNAZIONALI

Per informazioni: PROMOSALONS - ITALIA - Viale Teodorico, 19/2 - 20149 MILANO
Tel.: (02) 34.58.651/2/3 - Telex: 333448 PROSAL i